Example of a function constrained to a circle solved using Lagrange Multiplier

```
> restart:
> f := (x,y) -> x+y ;
```

$$f := (x, y) \rightarrow x + y \quad (1)$$

```
> h := (x,y) -> 1-x^2-y^2 ;
```

$$h := (x, y) \rightarrow 1 - x^2 - y^2 \quad (2)$$

Build the non linear system of the Lagrange multiplier

```
> gradf := [ diff(f(x,y),x),  diff(f(x,y),y) ] ;
```

$$gradf := [1, 1] \quad (3)$$

```
> gradh := [ diff(h(x,y),x),  diff(h(x,y),y) ] ;
```

$$gradh := [-2x, -2y] \quad (4)$$

```
> EQ1 := gradf[1] - lambda * gradh[1] = 0 ;
  EQ2 := gradf[2] - lambda * gradh[2] = 0 ;
  EQ3 := h(x,y) = 0 ;
```

$$EQ1 := 1 + 2\lambda x = 0$$

$$EQ2 := 1 + 2\lambda y = 0$$

$$EQ3 := 1 - x^2 - y^2 = 0 \quad (5)$$

```
> #solve({EQ1,EQ2,EQ3},{x,y,lambda} ) ; allvalues(%) ;
```

Solve the nonlinear system step by step

```
> SOLx := solve( EQ1, {x} ) ;
  SOLy := solve( EQ2, {y} ) ;
```

$$SOLx := \left\{x = -\frac{1}{2\lambda}\right\}$$

$$SOLy := \left\{y = -\frac{1}{2\lambda}\right\} \quad (6)$$

```
> EQlambda := subs(SOLy, subs( SOLx, EQ3 ) ) ;
```

$$EQlambda := 1 - \frac{1}{2\lambda^2} = 0 \quad (7)$$

```
> SOLlamda := solve( EQlambda, {lambda} ) ;
```

$$SOLlamda := \left\{\lambda = \frac{1}{2}\sqrt{2}\right\}, \left\{\lambda = -\frac{1}{2}\sqrt{2}\right\} \quad (8)$$

The two solutions are

```
> SOL1 := op(SOLlamda[1]), op(subs(SOLlamda[1], [op(SOLx),op(SOLy)] )
  ) ;
  SOL2 := op(SOLlamda[2]), op(subs(SOLlamda[2], [op(SOLx),op(SOLy)] )
  ) ;
```

$$SOL1 := \lambda = \frac{1}{2}\sqrt{2}, x = -\frac{1}{2}\sqrt{2}, y = -\frac{1}{2}\sqrt{2}$$

(9)

$$SOL2 := \lambda = -\frac{1}{2}\sqrt{2}, x = \frac{1}{2}\sqrt{2}, y = \frac{1}{2}\sqrt{2} \quad \textbf{(9)}$$

Check the solution

```
> subs(SOL1,[EQ1,EQ2,EQ3] ) ;
  subs(SOL2,[EQ1,EQ2,EQ3] ) ;
```

$$[0 = 0, 0 = 0, 0 = 0]$$

$$[0 = 0, 0 = 0, 0 = 0] \quad \textbf{(10)}$$

Find the space of vectors that are in the kernel of the contraint h(x,y)

```
> with(LinearAlgebra) :
> subs( SOL1, NullSpace(gradh) ) ;
```

$$\left\{ \begin{bmatrix} -1 \\ 1 \end{bmatrix} \right\} \quad \textbf{(11)}$$

```
> subs(SOL1,gradh[1]*alpha + gradh[2]*beta) ;
```

$$\sqrt{2}\,\alpha + \sqrt{2}\,\beta \quad \textbf{(12)}$$

Build the hessian

```
> hessf := <<diff(f(x,y),x,x),diff(f(x,y),x,y)>|<diff(f(x,y),x,y),
  diff(f(x,y),y,y)>> ;
```

$$hessf := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \quad \textbf{(13)}$$

```
> hessh := <<diff(h(x,y),x,x),diff(h(x,y),x,y)>|<diff(h(x,y),x,y),
  diff(h(x,y),y,y)>> ;
```

$$hessh := \begin{bmatrix} -2 & 0 \\ 0 & -2 \end{bmatrix} \quad \textbf{(14)}$$

```
> hessL := hessf - lambda * hessh ;
```

$$hessL := \begin{bmatrix} 2\lambda & 0 \\ 0 & 2\lambda \end{bmatrix} \quad \textbf{(15)}$$

Analyse the first solution

```
> hessL1 := subs( SOL1, hessL ) ;
```

$$hessL1 := \begin{bmatrix} \sqrt{2} & 0 \\ 0 & \sqrt{2} \end{bmatrix} \quad \textbf{(16)}$$

The vector in the kernel of grad h(x,y)

```
> z := alpha*<1,-1> ;
```

$$z := \begin{bmatrix} \alpha \\ -\alpha \end{bmatrix} \quad \textbf{(17)}$$

```
> Transpose(z).hessL1.z ;
```

$$2\sqrt{2}\,\alpha^2 \quad \textbf{(18)}$$

Analyse the second solution

```
> subs( SOL2, NullSpace(gradh) ) ;
```

$$\left\{\begin{bmatrix} -1 \\ 1 \end{bmatrix}\right\} \quad (19)$$

```
> hessL2 := subs( SOL2, hessL ) ;
```

$$hessL2 := \begin{bmatrix} -\sqrt{2} & 0 \\ 0 & -\sqrt{2} \end{bmatrix} \quad (20)$$

The vector in the kernel of grad h(x,y)

```
> z := alpha*<1,-1> ;
```

$$z := \begin{bmatrix} \alpha \\ -\alpha \end{bmatrix} \quad (21)$$

```
> Transpose(z).hessL2.z ;
```

$$-2\sqrt{2}\,\alpha^2 \quad (22)$$